CABIOS Vol.4. no.3. 1988
Page 407

SCATTER: a program to correct UV spectra recorded with an LKB Ultrospec interfaced to an Apple IIe for light scattering by particles or aggregates

Christiane Bergmann and Rolf Bergmann[1]

Conformational changes in nucleic acids can be estimated by measuring the peak absorbance in UV. When applying this method to intact viruses, we were confronted with the fact that our samples became turbid due to aggregation under the experimental conditions employed. However, a method already exists to measure the absorbance of the macromolecules in spite of light scattering (Englander and Epstein, 1957). The method is applicable to solutes which absorb UV but not visible light. Optical density (OD) by light scattering is a hyperbolic function of the wavelength. This function may be linearized by calculating the logarithms of both the OD and the wavelength. For wavelength values of visible light the OD is a result of scattering alone, and the resulting curve is a straight line. This line may be extrapolated to the UV part of the spectrum, giving the values for light scattering at these wavelengths. The difference between total OD and the value for scattering represents the intrinsic absorbance of the solute. This is illustrated in Figure 1. The spectra were recorded with an LKB Ultrospec photometer and the data stored on disk with an Apple IIe with the aid of software supplied by LKB.

The program SCATTER retrieves the data file chosen by the user and replots the spectrum with double logarithmic axes. The user is then asked to define the limits of the linear region of the plot, which is done by visual inspection on the screen. The program then calculates by least-squares fitting a straight line for the data points within the set limits. From the original OD data the wavelength of maximal OD is found, and for this wavelength the value of scattering is calculated and subtracted from the measured OD value. The user may choose whether to print the double logarithmic plot, the data for each wavelength or any combination of these. Graphic printing instructions are implemented for Grappler and Epson (APL) type interfaces. The program uses standard Applesoft instructions only and may easily be modified to include further calculations useful for a particular experimental set-up.

Acknowledgement

We thank R.Dernick for introducing the problem to us.

Heinrich Pette Institut für Experimentelle Virologie und Immunologie an der Universität Hamburg, Martinistrasse 52, D 2000 Hamburg 20 and [1]Universität Hamburg, Institut für Allgemeine Botanik, Ohnhorststrasse 18, D 2000 Hamburg 52, FRG



Fig. 1. (a) Sample spectrum as recorded by Ultrospec/Apple. (b) Double logarithmic replot by SCATTER. The measured OD is split into the part caused by light scattering (OD_{sc}) and the intrinsic absorption of the solute (A_i).

Reference

Englander,S.W. and Epstein,H.T. (1957) Optical methods for measuring nucleoprotein and nucleic acid concentrations. *Arch. Biochem. Biophys.*, **68**, 144–149.

```
100  REM   ************************************************************************
110  REM   The program calculates scatter from the visible part of a spectrum
120  REM   recorded with a LKB Ultrospec and stored on diskette. It finds the
130  REM   maximal absorbance in the UV part of the spectrum and corrects the
140  REM   peak value for scatter. The method of Englander & Epstein is used
150  REM    (Arch.Biochem.Biophys. 68 (1957) 144-149).
160  REM   ************************************************************************
165  REM  This version of scatter expects data format of ULTROSPEC I software
166  REM
170  REM   Input all ASCII
180  FOR X = 768 TO 793: READ Y
190  POKE X,Y
200  NEXT X
210  DATA    32,6,227,32,190,222,32,227,223,32,108,221,133,133,132,134,32,44,
213,200,32,233,227,76,154,218
220  HIMEM: 28671: LOMEM: 24576
230 D$ =  CHR$ (4)
240  DIM L(701),W(10),H(10)
250  PRINT : PRINT D$;"BLOAD PLOTDATA,A3072,D1"
260  POKE 232,0: POKE 233,12
270  SCALE= 1: ROT= 0
280  HOME
290  VTAB 18: PRINT "(For CATALOG of D2 press <Return>)"
300  VTAB 10
310  PRINT "Filename of Ultrospec data file": PRINT : PRINT "(disk in drive 2)"
320  VTAB 12: HTAB 19: PRINT "? ";
330  CALL 768,A$
340  IF  LEFT$ (A$,1) = "" THEN  GOTO 1840
350  POKE 43624,2: REM  Now drive 2 for input
360  ONERR  GOTO 280
370  PRINT : PRINT  CHR$ (4);"OPEN";A$
380  PRINT : PRINT  CHR$ (4);"READ";A$
390  INPUT T$
400  INPUT W1,W2,D
410 A = W2 - W1
420  RESTORE
430  FOR N = 1 TO A
440  INPUT L(N)
450  NEXT N
460  PRINT : PRINT  CHR$ (4);"CLOSE";A$
470  POKE 43624,1: REM  back to drive 1
480 E = .4342945:NS = 0:ODA = 0
490  REM  PLOT
500  HGR : GOSUB 1870
510  HCOLOR= 3
520  HPLOT 20,0 TO 20,170 TO 270,170
525 XM = 0
530  FOR I = W1 TO W2 - 3
540 S = I - W1 + 2
545  IF I > 620 THEN  GOTO 630: REM  limit screen output to 620nm
550  IF L(S) <  = 10 OR L(S + 1) <  = 10 OR L(S) > 2400 OR L(S + 1) > 2400 THEN
GOSUB 2230: GOTO 610: REM  discard out of range values for screen output
560 X1 =  LOG (I / 100) * E * 500 - 130
570 X2 =  LOG ((I + 1) / 100) * E * 500 - 130
580 Y1 = 170 -  LOG (L(S) / 10) * E * 70
590 Y2 = 170 -  LOG (L(S + 1) / 10) * E * 70
600  HPLOT X1,Y1 TO X2,Y2
610  NEXT I
615  REM
620  REM    Axes lettering
630  FOR M = 2 TO 6
640  DRAW 2 AT ( LOG (M) * E * 500 - 130),170
650  NEXT M
660 M = 1: GOSUB 830
670  FOR M = 5 TO 25 STEP 5
680  GOSUB 830
690  NEXT M
700  DRAW  ASC ( LEFT$ (A$,1)) AT 30,190
710  FOR S = 2 TO  LEN (A$)
```

```
720  DRAW  ASC ( MID$ (A$,S,1)): NEXT S
730  IF  LEN (T$) < 2 THEN  GOTO 780
740  DRAW 32: DRAW 47: DRAW 32
750  FOR S = 1 TO ( LEN (T$))
760  DRAW  ASC ( MID$ (T$,S,1))
770  NEXT S
780  DRAW 50 AT 2,12: DRAW 49 AT 2,33: DRAW 46 AT 0,54: DRAW 53: DRAW 46 AT 0,102:
DRAW 49
790  DRAW 79 AT 25,8: DRAW 68
800  DRAW 50 AT 10,180: DRAW 48: DRAW 48: DRAW 51 AT 98,180: DRAW 48: DRAW 48: DRAW
52 AT 161,180: DRAW 48: DRAW 48: DRAW 53 AT 209,180: DRAW 48: DRAW 48: DRAW 54 AT
249,180: DRAW 48: DRAW 48
810  DRAW 110 AT 232,189: DRAW 109
820  GOTO 840
830  DRAW 2 AT 20,(170 -  LOG (M * 10) * E * 70): RETURN
835  REM
840  REM  FIND LAMBDA MAX
850 PN = 0
860 WP = 260:HP = L(260 - W1): REM   if no peak found calculate for 260 nm
870  FOR I = 10 TO (A - 9)
880  IF L(I - 9) + 4 > L(I) THEN  NEXT : GOSUB 930: GOTO 935
890  IF L(I + 9) + 4 > L(I) THEN  NEXT : GOSUB 930: GOTO 935
900  IF L(I + 9) > L(I - 9) THEN  NEXT : GOSUB 930: GOTO 935
910 WP = I + W1:HP = L(I): GOSUB 930
920  GOTO 935
930 PN = PN + 1:W(PN) = WP:H(PN) = HP / 1000: RETURN
935  IF XM > 0 THEN  IF XM < 330 THEN  GOTO 2500
940  REM
950  REM    least squares fit
960  REM  define linear part of curve
970  REM       left border
980  POKE  - 16301,0: REM  mixed on
990  GOSUB 2360
1000  VTAB 21: PRINT "left border of linear part ": PRINT : PRINT "   <- J   K ->
/ ready: spacebar"
1010 XP = 125: REM  323 nm
1020  GOSUB 1110: GOSUB 1120
1030 LWL =  INT (100 * (10 ^ ((XP + 130) / 500)))
1040  REM       right border
1045  IF XM > 0 THEN  GOTO 2260
1050  VTAB 21: PRINT "right border of linear part ": PRINT : PRINT "   <- J   K ->
/ ready : spacebar"
1060 XP = 245: REM     562 nm
1070  GOSUB 1110: GOSUB 1120
1080 RWL =  INT (100 * (10 ^ ((XP + 130) / 500)))
1090  IF RWL > W2 THEN  GOTO 2440
1100  GOTO 1170
1110  FOR YP = 30 TO 130 STEP 10: XDRAW 108 AT XP,YP: NEXT YP: RETURN
1120  GET B$:B =  ASC (B$)
1130  IF B = 32 THEN  GOSUB 1110: RETURN
1140  IF B = 74 OR B = 8 THEN  GOSUB 1110:XP = XP - 4: GOSUB 1110: GOTO 1120
1150  IF B = 75 OR B = 21 THEN  GOSUB 1110:XP = XP + 4: GOSUB 1110: GOTO 1120
1160  GOTO 1120
1170  IF LWL > RWL THEN  GOTO 2410
1180  GOSUB 2360
1190  POKE  - 16302,0: REM  mixed off
1200  REM  Least-squares fitting
1210 S1 = 0:S2 = 0:S3 = 0:S4 = 0:S5 = 0
1220  REM   SUM X, X^2, Y, X*Y; X=lg WL, Y=lg OD
1230  FOR N = (LWL - W1) TO (RWL - W1)
1240  IF L(N) < 1 THEN L(N) = 1: GOTO 1300
1250 S1 = S1 +  LOG ((N + W1) / 100) * E * 500 - 130
1260 S2 = S2 + ( LOG ((N + W1) / 100) * E * 500 - 130) ^ 2
1270 S3 = S3 + (170 -  LOG (L(N) / 10) * E * 70)
1280 S5 = S5 + ( LOG ((N + W1) / 100) * E * 500 - 130) * (170 -  LOG (L(N) / 10) * E
* 70)
1290  XDRAW 98 AT 200,20: XDRAW 117: XDRAW 115: XDRAW 121
1300  NEXT N
1310  REM   equation formula Y=M*X+C
```

```
1320 N = RWL - LWL + 1
1330 M = (S5 - (S1 * S3 / N)) / (S2 - (S1 ^ 2 / N))
1340 C = (S3 - M * S1) / N
1350  DRAW 32 AT 200,20: DRAW 32: DRAW 32: DRAW 32
1360  GOSUB 1760: GOSUB 1870: GOSUB 1910: GOSUB 1910: GOSUB 1910
1370  REM    aggregation  OD  at lambda max
1380 XMAX =  LOG (W(1) / 100) * E * 500 - 130
1390 YAUSGL = M * XMAX + C
1400 ODA = (10 ^ ((170 - YAUSGL) / 70)) / 100
1410 ODA = ( INT (ODA * 10000)) / 10000
1420  REM   Print result
1430  GOSUB 1440: GOTO 1510
1440  TEXT : HOME
1450  PRINT "         lambda max = ";W(1);" nm"
1460  PRINT "         in spectrum ";A$
1470  PRINT : PRINT "       OD in spectrum  = ";H(1)
1480  PRINT : PRINT "       OD by scatter   = ";ODA
1490  PRINT : PRINT "       corrected OD    = ";H(1) - ODA
1500  RETURN
1505  REM
1510  REM   PRINT
1520  VTAB 10: PRINT "_____________________________________"
1530  PRINT
1540  PRINT "   print to paper:": PRINT
1550  PRINT "       result      <1>"
1560  PRINT "        data       <2>"
1570  PRINT " logarithmic plot <3>"
1580  PRINT "       1 + 2       <4>"
1590  PRINT "       1 + 3       <5>"
1600  PRINT "       2 + 3       <6>"
1610  PRINT "       1 + 2 + 3   <7>": PRINT
1620  PRINT "    next spectrum  <0>"
1630  PRINT "       Exit        <E>"
1640  GET PR$
1650  IF PR$ =  CHR$ (69) OR PR$ =  CHR$ (101) THEN  HOME : END
1655  IF PR$ =  CHR$ (13) AND XM > 0 THEN XP =  LOG (XM / 100) * E * 500 - 130:
GOSUB 1870: GOTO 1080
1660 PR =  VAL (PR$)
1670  IF PR = 0 THEN  HOME : GOTO 300
1680  IF PR = 1 THEN  GOSUB 2000: PRINT : PRINT : GOSUB 1450: PRINT : PRINT : GOSUB
2030: GOTO 1430
1690  IF PR = 2 THEN  GOSUB 1440: GOSUB 1920: GOSUB 2000: PRINT : PRINT : GOSUB
2070: PRINT : PRINT : GOSUB 2030: GOTO 1430
1700  IF PR = 3 THEN  GOSUB 1440: GOSUB 1950: GOSUB 2000: GOTO 1830
1710  IF PR = 4 THEN  GOSUB 1440: GOSUB 1920: GOSUB 2000: PRINT : PRINT : GOSUB
2070: PRINT : PRINT : GOSUB 1450: PRINT : PRINT : GOSUB 2030: GOTO 1430
1720  IF PR = 5 THEN  GOSUB 1440: GOSUB 1950: GOSUB 2000: GOSUB 1450: GOTO 1830
1730  IF PR = 6 THEN  GOSUB 1440: GOSUB 1920: VTAB 20: GOSUB 1960: GOSUB 2000: GOSUB
2070: GOTO 1830
1740  IF PR = 7 THEN  GOSUB 1440: GOSUB 1920: VTAB 20: GOSUB 1960: GOSUB 2000: GOSUB
1450: PRINT : PRINT : GOSUB 2070: GOTO 1830
1750  GOTO 1640
1755  REM
1760  REM   Plot calculated curve
1770 XL =  INT ( LOG (W1 / 100) * E * 500 - 130)
1780 XR =  INT ( LOG (W2 / 100) * E * 500 - 130)
1790  FOR I = XL TO XR STEP 3
1800  IF (M * I + C) < 0 OR (M * I + C) > 170 THEN  GOTO 1820
1810  XDRAW 17 AT I,(M * I + C)
1820  NEXT I: RETURN
1830  ON PI GOSUB 2050,2060: GOSUB 2030: GOTO 1430
1840  HOME : PRINT : PRINT  CHR$ (4);"CATALOG,D2"
1850  PRINT : PRINT " press <Return> to go on ": CALL 768,R$
1860  HOME : GOTO 300
1870  POKE  - 16304,0: REM  graphics screen on
1880  POKE  - 16297,0
1890  POKE  - 16302,0
1900  RETURN
1910  FOR I = 1 TO 1000: NEXT : RETURN
```

```
1920  VTAB 16
1930  PRINT A;" data points exist.          ": PRINT : PRINT "Enter step size for
printout :"
1940  INPUT DD: RETURN
1950  VTAB 16
1960  PRINT "   Enter type of printer interface :": PRINT
1970  PRINT "         Grappler     <1>"
1980  PRINT "         Epson (APL)  <2>"
1990  GET PI: RETURN
2000  PRINT : PRINT D$;"PR#1"
2010  PRINT : PRINT  CHR$ (9);"80N"
2020  RETURN
2030  PRINT : PRINT  CHR$ (12): PRINT D$;"PR#0"
2040  RETURN
2050  PRINT : PRINT  CHR$ (9);"GDR": RETURN
2060  PRINT : POKE 1273,65: PRINT  CHR$ (17): RETURN
2070  REM    Data printout
2080  PRINT "       Data for spectrum  ";A$
2090  PRINT : PRINT "       Wavel.(nm) OD      *  Log(WL/100)    Log(OD*100)": PRINT
2100  FOR I = W1 TO W2 STEP DD
2110 S = I - W1 + 1
2120  IF L(S) <  = 0 THEN  GOTO 2210
2130 OD$ =  STR$ (L(S))
2140  IF  LEN (OD$) = 1 THEN OD$ = "0.00" + OD$
2150  IF  LEN (OD$) = 2 THEN OD$ = "0.0" + OD$
2160  IF  LEN (OD$) = 3 THEN OD$ = "0." + OD$
2170  IF  LEN (OD$) = 4 THEN OD$ =  LEFT$ (OD$,1) + "." +  RIGHT$ (OD$,3)
2180 LGWL$ =  LEFT$ ( STR$ ( LOG (I / 100) * E),6)
2190  PRINT "       ";
2200  PRINT "    ";I;"   ";OD$;"    *    ";LGWL$;"      "; LOG ( ABS (L(S) / 10)) *
E
2210  NEXT I
2220  RETURN
2230  IF L(S) > 10 THEN  RETURN
2240 NS = NS + 1: REM   count low OD values
2245  IF L(S) < 4 THEN XM = W1 + S: GOTO 630
2250  RETURN
2260  TEXT : HOME : VTAB 4
2270  PRINT "OD values above ";XM;" nm in this": PRINT "spectrum are too low to
calculate": PRINT "scatter. Now what:"
2280  PRINT
2290  PRINT "calculate anyway from left border"
2300  PRINT "up to ";XM;" nm       <Return>"
2310  PRINT
2330  PRINT "    next spectrum       <0>"
2340  PRINT "       Exit             <E>"
2350  GOTO 1640
2360  FOR M = 3 TO 6
2370  XDRAW 2 AT ( LOG (M) * E * 500 - 130),15
2380  NEXT M
2390  XDRAW 51 AT 98,7: XDRAW 48: XDRAW 48: XDRAW 52 AT 161,7: XDRAW 48: XDRAW 48:
XDRAW 53 AT 209,7: XDRAW 48: XDRAW 48: XDRAW 54 AT 249,7: XDRAW 48: XDRAW 48
2400  RETURN
2410  PRINT : PRINT : VTAB 21: PRINT "The right border is to the left of": PRINT
"the left border - try again !"
2420  GOSUB 1910: GOSUB 1910: GOSUB 1910: PRINT : PRINT : GOTO 1000
2440  VTAB 24: PRINT "No data so far - move border to the left"
2450  PRINT  CHR$ (7)
2460  GOSUB 1910: GOSUB 1910: GOSUB 1910
2470  PRINT : PRINT : PRINT : GOTO 1050
2500  REM  no scatter found
2510  TEXT : HOME : PRINT
2520  PRINT "No scatter found in spectrum"
2530  PRINT : PRINT A$
2535  PRINT : PRINT
2540  PRINT "lambda max = ";W(1);" nm"
2550  PRINT "        OD = ";H(1);
2560  GOTO 1520
```

Abb. A 1: VP2



Abb. A 2: Seitenansicht der antigenen Region 1



Abb. A 3: Aufsicht der antigenen Region 2



Abb. A 4: Ansicht der antigenen Region 3



Abb. A 5: Die Lage der Aminosäuren A2240 und P2245

Danksagung

Allen, die mir diese Arbeit möglich gemacht haben, sei herzlich gedankt.

Herrn Prof. Dr. W. Abel danke ich dafür, daß ich mit seiner Hilfe ein Promotionsstudium in Hamburg beginnen durfte.

Besonders Herrn Prof. Dr. R. Dernick danke ich für die Vergabe des Themas meiner Arbeit, der Bereitstellung von Arbeitsplatz und Material in der Abteilung "Biochemie der Viren" am Heinrich-Pette-Institut für Experimentelle Virologie und Immunologie an der Universität Hamburg.

Mein bester Dank gilt Herrn Heukeshoven für seine umfassende Hilfe bei der Aufarbeitung von Virusproteinen und Viruspeptiden, sowie seiner Ausdauer beim Ausfeilen von Methoden. Herrn Wiegers danke ich für das Überlassen von Mausaszites, Frau Hilbrig für das Heraussuchen von Virusproben und Seren, die dankenswertereweise von der Abteilung hergestellt waren, Frau Uhlig besonders für den Antikörper 7J6, Frau Heigl für das Abwiegen von DAB, Frau Gardemin für die Bestellung von Literatur.

Ein großer Dank gebührt der mikrobiologischen Abteilung des Botanischen Institutes in Hamburg für die Benutzung ihrer Computer; besonders meinem Mann, der mich in verschiedenste Programme einarbeitete, mir bei der Literaturverwaltung und bei der Erarbeitung eines eigenen Programms half. Dank sei auch dem Institut für Physikalische Chemie der Universität Hamburg für die Benutzung der VAX.

Bester Dank gebührt L. Menedez-Arias und R. Rodriguez, sowie O. Gotoh und der Mikrobiologie, die mir ihre Programme kostenlos zur Verfügung stellten.

Meiner Tochter Julia danke ich für die Erlärung der Matrixrechnung, die zum Erstellen der Strukturbilder nötig war.

Ich danke Frau Kuck, Frau Hilbrig, Julia und Mario, die mich immer wieder aufgemuntert haben, und Tom, der mir immer wieder die Unwichtigkeit meiner Arbeit vor Augen führte.

Lebenslauf

Christiane Bergmann geb. Lengelsen

Geburt:	17.12.49 in Altena/Westf.
Vater:	Friedrich Wilhelm Lengelsen (Geschäftsführer)
Mutter:	Rita Lengelsen geb. Beichel (Apothekerin)
Geschwister:	2 jüngere Brüder

geheiratet:	1975 Dr.rer.nat. Rolf Bergmann (Chemiker, Mikrobiologe)
Kinder:	Julia Fee Bergmann geb. 1975
	Mario Lars Bergmann geb. 1979
	Tom Alexander Bergmann geb. 1982

1969: Abitur in Altena (neusprachliches Mädchengym.)

ab 1969: Studium der Biologie in Bochum

ab 1972: Betreuung verschiedener mikrobiologischer Praktika

1974: Biologiediplom (Schwerpunkt: Mikrobiologie, Biochemie)

Thema der Diplomarbeit: Untersuchungen von ribosomaler RNS während Wuchs und Sporulation von Bacillus subtilis (Spores VI)

1977: Umzug nach Hamburg

ab 1977 ehrenamtliche Beratungen: Stillen, Ernährung,, Umweltschutz

Veröffentlichungen zu : Vitamin D, Schadstoffe in Muttermilch

1977: FEBS Advanced Course No.48 in Moskau und Pushchino

1978: 12th FEBS Meeting Dresden

ab 1985: Promotionsstudium an der Universität Hamburg



Hiermit versichere ich an Eides statt, daß zur Durchführung und Abfassung der vorliegenden Dissertation keine anderen als die angegebenen Hilfsmittel verwendet wurden. Die Arbeit wurde an keiner anderen Universität eingereicht.

Hamburg, den 1. 4. 94

C. Berg